Komparative Statik

Gegeben:
Produktionsfunkion

```
In[1]:= y = l^a k^b;
```

Preise: p, w, r

a) Optimierungsproblem, Entscheidungsvariablen und Parameter

Entscheidungsvariablen: l, k

```
In[2]:= ev = {l, k};
```

Parameter: p, w, r

```
In[3]:= para = {p, w, r};
```

Optimierungsproblem:

```
gewinn = p * y - w * l - r * k;

Max[gewinn, {l, k}]
```

b) Bedingungen erster Ordnung

Leiten den Gewinn nach den Entscheidungsvariablen ab:

```
In[5]:= (abl = D[gewinn, {ev}]) // TableForm
```

Out[5]//TableForm=

$a\,k^b\,l^{-1+a}\,p - w$
$b\,k^{-1+b}\,l^a\,p - r$

Setzen Ableitungen gleich Null

```
In[6]:= (foc = ((# == 0) & /@ abl)) // TableForm
```

Out[6]//TableForm=

$a\,k^b\,l^{-1+a}\,p - w == 0$
$b\,k^{-1+b}\,l^a\,p - r == 0$

c) Komparative Statik

Behandeln nun die Entscheidungsvariablen als Abhängige der Parameter.
Bspw. als Abhängige vom Lohnsatz w

```
In[7]:= foc /. {k → k[w], l → l[w]} // TableForm
```

Out[7]//TableForm=

$-w + a\,p\,k[w]^b\,l[w]^{-1+a} == 0$
$-r + b\,p\,k[w]^{-1+b}\,l[w]^a == 0$

Leiten nach dem Lohnsatz w ab:

In[10]:=
```
(eqs = D[foc /. {k → k[w], l → l[w]}, w]) // TableForm
```

Out[10]//TableForm=

$$-1 + a\, b\, p\, k[w]^{-1+b}\, l[w]^{-1+a}\, k'[w] + (-1 + a)\, a\, p\, k[w]^{b}\, l[w]^{-2+a}\, l'[w] == 0$$
$$(-1 + b)\, b\, p\, k[w]^{-2+b}\, l[w]^{a}\, k'[w] + a\, b\, p\, k[w]^{-1+b}\, l[w]^{-1+a}\, l'[w] == 0$$

Haben also ein lineares Gleichungssysten in den Ableitungen: k′[w], l′[w]

Lösen diese nach den Ableitungen auf:

In[11]:=
```
Solve[eqs, {k'[w], l'[w]}][[1]] // TableForm
```

Out[11]//TableForm=

$$k'[w] \to \frac{k[w]^{1-b}\, l[w]^{1-a}}{(-1+a+b)\, p}$$
$$l'[w] \to -\frac{(-1+b)\, k[w]^{-b}\, l[w]^{2-a}}{a\, (-1+a+b)\, p}$$

Untersuchung nun die Änderungen der Kapital- und Arbeitsnachfrage in Abhängigkeit vom Lohnsatz.

Ansatz:

```
dk =  k'[w] * dw
dl = l'[w] * dw
```

Vorzeichen von k′[w]?

Da a+b < 1, ist (-1+a+b) < 0.

Also k′[w] < 0

Bei l′[w]:

Da 0 < b < 1, ist (-1+b) <0

Also l′[w] < 0

Also:

Wenn dw > 0, dann dk, dl < 0

Wenn dw < 0, dann dk, dl > 0

Nun für alle Parameter:

In[12]:=
```
compStat[variable_] := Solve[D[foc /. {k → k[variable], l → l[variable]}, variable],
    {k'[variable], l'[variable]}][[1]]
Table [compStat[variable], {variable, para}] // TableForm
```

Out[13]//TableForm=

$k'[p] \to -\frac{k[p]}{(-1+a+b)\, p}$	$l'[p] \to -\frac{l[p]}{(-1+a+b)\, p}$
$k'[w] \to \frac{k[w]^{1-b}\, l[w]^{1-a}}{(-1+a+b)\, p}$	$l'[w] \to -\frac{(-1+b)\, k[w]^{-b}\, l[w]^{2-a}}{a\, (-1+a+b)\, p}$
$k'[r] \to -\frac{(-1+a)\, k[r]^{2-b}\, l[r]^{-a}}{b\, (-1+a+b)\, p}$	$l'[r] \to \frac{k[r]^{1-b}\, l[r]^{1-a}}{(-1+a+b)\, p}$

Vorzeichen der Ableitungen in obiger Reihenfolge

Echt Positiv?

In[14]:=
```
assum[variable_] :=
 k[variable] > 0 && l[variable] > 0 && 0 < a && 0 < b && a + b < 1 && p > 0 && w > 0 && r > 0
Table[Assuming[assum[variable],
   Simplify[{k′[variable] > 0, l′[variable] > 0} /. compStat[variable]]],
  {variable, para}] // TableForm
```

Out[15]//TableForm=
```
True   True
False  False
False  False
```

Echt Negativ?

In[17]:=
```
Table[Assuming[assum[variable],
   Simplify[{k′[variable] < 0, l′[variable] < 0} /. compStat[variable]]],
  {variable, para}] // TableForm
```

Out[17]//TableForm=
```
False  False
True   True
True   True
```

Integration

a) Konsumenten- u. Produzentenrente

Marktgleichgewicht

Gegeben:

In[18]:=
```
xd = a - b * p;
xs = c * p;
```

Setzen Angebot u. Nachfrage gleich:

In[20]:=
```
eq = (xd == xs)
```

Out[20]= $a - b\,p == c\,p$

Lösen nach Preis auf:

In[21]:=
```
sol = Solve[eq, p][[1]]
```

Out[21]= $\left\{p \to \frac{a}{b + c}\right\}$

In[22]:=
```
pgg = p /. sol;
```

Setzen in Angebot ein:

In[23]:=
```
x → (xgg = xs /. sol)
```

Out[23]= $x \to \frac{a\,c}{b + c}$

Wie sehen Angebot und Nachfrage aus?
(Bspw. für a = 1, b = 1, c=1)

In[24]:=
```
subs = {a → 1, b → 1, c → 1};
Show[Plot[Evaluate[{xs, xd} /. {a → 1, b → 1, c → 1}], {p, 0, 1},
   PlotLegends → {Angebot, Nachfrage}, AxesLabel → {Preis, Menge}],
  Plot[Evaluate[{xs, xd} /. subs], {p, 0, (xgg /. subs)}, Filling → (xgg /. subs)]]
```

Out[25]=



Konsumentenrente ist also die Fläche unter der Nachfrage bis zur Gleichgewichten Menge und dem gleichgewichtigen Preis.
Integrieren die Differenz zwischen Nachfrage und gleichgewichtiger Menge:

In[26]:= xd - xgg

Out[26]= $a - \frac{a\,c}{b+c} - b\,p$

Etwas zusammengefasster:

In[27]:= (xd - xgg) // Simplify

Out[27]= $\frac{a\,b}{b+c} - b\,p$

Integrieren nun bis zum gleichgewichtigen Preis:

$$\boxed{\begin{array}{l} \int (xd - xgg)\, dp \\ \int \left(\frac{a\,b}{b+c} - b\,p\right) dp \\ \int \left(\frac{a\,b}{b+c}\right) dp - \int (b\,p)\, dp \\ \frac{a\,b}{b+c} * p - b * (1/2) * p^2 \\ \frac{a\,b\,p}{b+c} - \frac{b\,p^2}{2} \end{array}}$$

In[37]:= $\frac{a\,b}{b+c} * p - b * (1/2) * p^2$

Out[37]= $\frac{a\,b\,p}{b+c} - \frac{b\,p^2}{2}$

In den Grenzen von 0 bis zum gleichgewichtigen Preis pgg.

In[39]:= $\left(\left(\frac{a\,b}{b+c} * p - b * (1/2) * p^2\right) /.\ \{p \to pgg\}\right) - \left(\left(\frac{a\,b}{b+c} * p - b * (1/2) * p^2\right) /.\ \{p \to 0\}\right)$

Out[39]= $\boxed{\frac{a^2\,b}{2\,(b+c)^2}}$

In[31]:=
```
Integrate[xd - xgg, {p, 0, pgg}]
```

Out[31]= $-\frac{a^2 b}{2 (b+c)^2}+\frac{a \left(a-\frac{a c}{b+c}\right)}{b+c}$

Vereinfachter:

In[33]:=
```
int = Integrate[(xd - xgg), {p, 0, pgg}] // Simplify
```

Out[33]= $\frac{a^2 b}{2 (b+c)^2}$

Für obiges Beispiel haben wir dann für die Konsumentenrente:

In[ ]:=
```
int /. subs
```

Out[ ]= $\boxed{\frac{1}{8}}$

Analog für Produzentenrente.

Integrieren Differenz zwischen gleichgewichtiger Menge und Angebot

In[34]:=
```
xs - xgg
```

Out[34]= $-\frac{a c}{b+c}+c p$

Integrieren bis zum gleichgewichten Preis

$\int (xgg - xs)\, dp$

In[35]:=
```
int = Integrate[xgg - xs, {p, 0, pgg}]
```

Out[35]= $\frac{a^2 c}{2 (b+c)^2}$

Haben für obiges Beispiel:

In[36]:=
```
int /. subs
```

Out[36]= $\boxed{\frac{1}{8}}$

b) Differentialgleichungen

Gegeben ist folgende Differentialgleichung:

In[ ]:=
```
D[c[t], t] / c[t] == g
```

Out[ ]= $\frac{c'[t]}{c[t]} == g$

Integrieren beide Seiten

In[40]:= $\int \frac{c'[t]}{c[t]}\, dt == \int g\, dt$

```
Log[c[t]] == g t + const
```

Etwas ausführlicher:

Schreiben die Ableitung in der Leibnizschen Schreibweise:

$c'[t] = dc[t] / dt$

Setzen in das Integral ein:

$$\int \frac{dc[t]/dt}{c[t]} \, dt = \,..$$

kürzen formal das *d*t raus

$$.. = \int \frac{dc[t]}{c[t]} = \,..$$

Haben also ein Integral nach c[t]. Behandeln dies wie eine normale Integrationsvariable (ignorieren die Abhängigkeit von t) und Integrieren.

$$.. = \int \frac{dc}{c} \left(= \int \frac{1}{c} \, dc \right) = \mathrm{Log}[c]$$

Zurück zum Ausgangspunkt. Haben nun also als Gleichung

In[43]:= `res = (Log[c[t]] == g * t + const)`

Out[43]= `Log[c[t]] == const + g t`

Wenden auf beide Seiten die Exponentialfunktion an

In[42]:= `Exp /@ res`

Out[42]= $c[t] == e^{\text{const}+g\,t}$

$c[t] == e^{\text{const}+g\,t}$

Anders geschrieben:

`c[t] = Exp[const + g * t] = Exp[const] * Exp[g * t]`

Bestimmen nun die Integrationskonstante const, bzw. deren Exponentialwert, durch einsetzen von t = 0

`c[0] = Exp[const] * Exp[0] = Exp[const]`

Nehmen an c[0] sei bekannt. Hätten damit unser Funktion c[t] folgendermaßen bestimmt:

`c[t] = Exp[const + gt] = Exp[const] * Exp[g * t] = c[0] * Exp[gt]`
`c[t] = c[0] * Exp[gt]`

Falls const direkt von Interesse muss auf c[0]=Exp[const] der Logarithmus angewandt werden. Hätten dann:

`Log[c[0]] = const`

Kämen nach einsetzen dann auf diesselbe Lösung.

`c[t] = Exp[const] * Exp[g * t] = Exp[Log[c[0]]] * Exp[g * t] = c[0] * Exp[g * t]`
`c[t] = c[0] * Exp[g * t]`